
100 Excel Formulas
Ready-to-edit formulas for the jobs people do most, in ten groups.

Change the ranges to fit your data. A tag marks formulas that need Microsoft 365 or a newer Excel version. Most
also work in Google Sheets, though argument details can differ.

Sum

What you need Formula

Add a range =SUM(B2:B50)

Add with one condition =SUMIF(A2:A50,"East",C2:C50)

Add with several conditions =SUMIFS(C2:C50,A2:A50,"East",B2:B50,"Feb")

Add between two dates =SUMIFS(C2:C50,A2:A50,">="&F1,A2:A50,"<="&F2)

Add visible rows only =SUBTOTAL(109,C2:C50)

Running total =SUM($C$2:C2)

Quantity times price, then add =SUMPRODUCT(B2:B50,C2:C50)

Add values above a limit =SUMIF(C2:C50,">1000")

Add the three largest =SUM(LARGE(C2:C50,{1,2,3}))

Add while ignoring errors (Excel) Newer Excel =AGGREGATE(9,6,C2:C50)

Count

What you need Formula

Count numbers =COUNT(B2:B50)

Count non-empty cells =COUNTA(A2:A50)

Count blank cells =COUNTBLANK(A2:A50)

Count with one condition =COUNTIF(A2:A50,"East")

Count with several conditions =COUNTIFS(A2:A50,"East",C2:C50,">1000")

Count text that contains a word =COUNTIF(A2:A50,"*east*")

Count unique values, no blanks =SUMPRODUCT(1/COUNTIF(A2:A50,A2:A50))

Count between two numbers =COUNTIFS(C2:C50,">=100",C2:C50,"<=500")

Count dates in one month =COUNTIFS(A2:A50,">="&DATE(2026,1,1),A2:A50,"
<"&DATE(2026,2,1))

Test for a repeated value =COUNTIF($A$2:$A$50,A2)>1

Look up

What you need Formula

Modern lookup Newer Excel =XLOOKUP(F2,A2:A50,C2:C50,"Not found")

Classic exact lookup =VLOOKUP(F2,A2:C50,3,FALSE)

What you need Formula

Lookup that works in every version =INDEX(C2:C50,MATCH(F2,A2:A50,0))

Look up by row and column =INDEX(B2:E50,MATCH(G1,A2:A50,0),MATCH(G2,B1:E1,0))

Return a column to the left =INDEX(A2:A50,MATCH(F2,C2:C50,0))

Find the last match Newer Excel =XLOOKUP(F2,A2:A50,C2:C50,,0,-1)

Find a rate from a tier table Newer Excel =XLOOKUP(F2,A2:A5,B2:B5,,-1)

Look up with two conditions =INDEX(C2:C50,MATCH(1,INDEX((A2:A50=F1)*(B2:B50=F2),0),0))

Look up part of a text Newer Excel =XLOOKUP("*"&F2&"*",A2:A50,C2:C50,"Not found",2)

Show a message for a missing value =IFNA(VLOOKUP(F2,A2:C50,3,FALSE),"Not listed")

Logic

What you need Formula

Choose between two results =IF(B2>=100,"Pass","Review")

Three outcomes with nested IF =IF(B2>=90,"A",IF(B2>=75,"B","C"))

Several tests Newer Excel =IFS(B2>=90,"A",B2>=75,"B",TRUE,"C")

All tests must pass =IF(AND(B2>0,C2>0),"OK","Check")

Any test can pass =IF(OR(A2="East",A2="West"),"Yes","No")

Match a list of values Newer Excel =SWITCH(A2,"E","East","W","West","Other")

Replace any error =IFERROR(A2/B2,0)

Replace only #N/A =IFNA(MATCH(F2,A2:A50,0),0)

Flag a blank cell =IF(A2="","Missing","OK")

Test a number range =IF(AND(B2>=10,B2<=20),"In range","Out of range")

Text

What you need Formula

Join with a separator Newer Excel =TEXTJOIN(", ",TRUE,A2:A9)

Join first and last name =A2&" "&B2

First three characters =LEFT(A2,3)

Last four characters =RIGHT(A2,4)

Characters from the middle =MID(A2,5,2)

Remove extra spaces =TRIM(A2)

Capitalize each word =PROPER(A2)

Replace one word with another =SUBSTITUTE(A2,"Ltd","Limited")

Get the first word =LEFT(A2,FIND(" ",A2)-1)

Format a number as text =TEXT(B2,"#,##0.00")

Dates

What you need Formula

Today =TODAY()

What you need Formula

Last day of the month =EOMONTH(A2,0)

First day of the month =A2-DAY(A2)+1

Add three months =EDATE(A2,3)

Complete years between dates =DATEDIF(A2,TODAY(),"Y")

Workdays between dates =NETWORKDAYS(A2,B2)

Date after 10 workdays =WORKDAY(A2,10)

Day name =TEXT(A2,"dddd")

Calendar quarter =ROUNDUP(MONTH(A2)/3,0)

Year and month label =TEXT(A2,"yyyy-mm")

Math and percentages

What you need Formula

Round to two decimals =ROUND(A2,2)

Round up to a whole number =ROUNDUP(A2,0)

Round to the nearest 5 =MROUND(A2,5)

Percent of total =B2/SUM($B$2:$B$50)

Percent change =(C2-B2)/B2

Profit margin =(B2-C2)/B2

Average =AVERAGE(B2:B50)

Average with conditions =AVERAGEIFS(C2:C50,A2:A50,"East")

Largest value =MAX(B2:B50)

Smallest value =MIN(B2:B50)

Rank and statistics

What you need Formula

Rank in a list =RANK.EQ(B2,$B$2:$B$50)

Second largest =LARGE(B2:B50,2)

Second smallest =SMALL(B2:B50,2)

Middle value =MEDIAN(B2:B50)

Most common number =MODE.SNGL(B2:B50)

Sample standard deviation =STDEV.S(B2:B50)

90th percentile =PERCENTILE.INC(B2:B50,0.9)

Correlation of two columns =CORREL(B2:B50,C2:C50)

Linear forecast =FORECAST.LINEAR(F2,C2:C50,B2:B50)

Upper quartile =QUARTILE.INC(B2:B50,3)

Dynamic arrays

What you need Formula

Filter by a condition Newer Excel =FILTER(A2:C50,B2:B50="East")

Filter by two conditions (Excel) Newer Excel =FILTER(A2:C50,(B2:B50="East")*(C2:C50>1000))

Sort by a column Newer Excel =SORT(A2:C50,3,-1)

Sort by another column (Excel) Newer Excel =SORTBY(A2:A50,C2:C50,-1)

Unique list Newer Excel =UNIQUE(A2:A50)

Sorted unique list Newer Excel =SORT(UNIQUE(A2:A50))

Top five rows (Excel) Newer Excel =TAKE(SORT(A2:C50,3,-1),5)

Numbers 1 to 10 Newer Excel =SEQUENCE(10)

Split text into columns (Excel) Newer Excel =TEXTSPLIT(A2," ")

Name part of a formula Newer Excel =LET(t,SUM(C2:C50),C2/t)

Clean and check

What you need Formula

Remove spaces and hidden characters =TRIM(CLEAN(A2))

Turn text into a number =VALUE(TRIM(A2))

Get an email domain =MID(A2,FIND("@",A2)+1,100)

Check for a real number =ISNUMBER(B2)

Mark repeated values =IF(COUNTIF($A$2:$A$50,A2)>1,"Duplicate","Unique")

Check a code length =LEN(A2)=10

Keep leading zeros =TEXT(A2,"00000")

Remove commas =SUBSTITUTE(A2,",","")

Change to capitals =UPPER(A2)

Replace a blank with zero =IF(A2="",0,A2)

Every function, free templates and practical AI guides for Excel and Google Sheets. www.excelintellect.com

